# Maintainer: Alexander Epaneshnikov <alex19ep@archlinux.org>
# Contributor: Brad Fanella <cesura@archlinux.org>
# Contributor: Martin Wimpress <code@flexion.org>

pkgname=mate-system-monitor
pkgver=1.26.1
pkgrel=1
pkgdesc="A system monitor for MATE"
url="https://mate-desktop.org"
arch=('loong64' 'x86_64')
license=('GPL')
depends=('gtkmm3' 'libgtop' 'libwnck3' 'gettext' 'polkit')
makedepends=('itstool' 'python')
conflicts=('mate-system-monitor-gtk3')
replaces=('mate-system-monitor-gtk3')
groups=('mate-extra')
source=("https://pub.mate-desktop.org/releases/${pkgver%.*}/${pkgname}-${pkgver}.tar.xz")
sha256sums=('1eb5fb9b6cb4a8ad830b26e847a9bbd01d568aabd383c628ee9f08409b8928e7')

build() {
	cd "${pkgname}-${pkgver}"
	./configure \
	            --prefix=/usr \
	           --libexecdir=/usr/lib/${pkgname} \
	           --localstatedir=/var \
	           --enable-systemd
	make
}

package() {
	cd "${pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}" install
}
